Credit Default Swap with Mathematica 10

Credit Default Swap which is better known as **CDS** is a popular derivative contract which enables counterparties to manage and control credit exposure to a reference credit entity. Under the contract, one side – protection buyer (seller) pays (receives) premium in return for credit loss compensation that is received (paid) when the reference entity defaults. In essence, the CDS is an insurance contract against reference entity default. Over the years, CDS has evolved in many directions and structural variations were proposed to adapt the contract to particular market needs. For example, the reference obligations can be single entity, index, baskets of few names or larger pools.

CDS is a basic building block for many other derivatives contracts and methods. In this way, it is closely linked to CVA and DVA. Therefore, it is natural to extend this exposition to provide a complete credit-related coverage within **Mathematica 10**, which again proves its competence and nice flexibility to handle this task as efficient as previously.

The mechanics of standard CDS is quite simple – there are two legs (premium and protection) where one party pays premium and receives protection or vice versa. Today, the contract is fairly standardised with quarterly payment frequency on the premium leg. The value of the contract is simply the difference between both legs.

Premium leg: = $c * N * \sum_1^T Exp[-(z+h) * i]dt$

Protection leg:= $(1-R) * N * h * \sum_1^T Exp[-(z+h) * i]dt$

Where c is the premium, N is the nominal amount, z is the risk-free zero –coupon rate, h is the hazard rate (default intensity), R is the recovery rate and dt is the year fraction – generally ¼.

Premium leg is essentially the premium amount discounted by risky annuity and the protection leg is default probability-adjusted loss amount discounted by the same annuity.

To value the CDS, one essentially needs (i) zero-coupon rates and (ii) hazard rates term structure. If we assume that the (i) is given to us, then the solution reduces to determining the hazard rates. These can be obtained from the term structure of the CDS rates quoted in the market by inverting the CDS valuation formula.

```
deq = c * Sum[f * Exp[-(r + h) * i], {i, f, T, f}] -
   (1 - R) * h * Sum[f * Exp[-r * i] * Exp[-h * i], {i, f, T, f}]
```

$$\frac{c\, e^{f h+f r}\left(e^{f(-h-r)}-e^{(-h-r)\left(f+f\left(1+\text{Floor}\left[\frac{-f+T}{f}\right]\right)\right)}\right) f}{-1+e^{f h+f r}}+\frac{e^{-h\left(f+f\left(1+\text{Floor}\left[\frac{-f+T}{f}\right]\right)\right)-r\left(f+f\left(1+\text{Floor}\left[\frac{-f+T}{f}\right]\right)\right)}\left(e^{f h+f r}-e^{h\left(f+f\left(1+\text{Floor}\left[\frac{-f+T}{f}\right]\right)\right)+r\left(f+f\left(1+\text{Floor}\left[\frac{-f+T}{f}\right]\right)\right)}\right) f\, h\,(1-R)}{-1+e^{f h+f r}}$$

All we need is solve this equation for **h**:

```
rsln = Solve[deq == 0, h] // Quiet
```

$$\left\{\left\{h \to -\frac{c}{-1+R}\right\}\right\}$$

The solution is surprisingly simple. The hazard rate is the ratio of quoted CDS (for a given maturity) divided by the loss fraction. This enables us to build the entire term structure of hazard rates quickly from the available CDS rates.

Assume we are given the following market data with $R = 40\%$ and $dt = 1/4$

```
r = {0.0105, 0.0115, 0.0123, 0.0128, 0.0136, 0.0139, 0.0145};
c = {0.005, 0.007, 0.01, 0.0125, 0.0132, 0.0148, 0.0155};
t = {0.25, 0.5, 1, 2, 3, 4, 5};
{R, f} = {0.4, 1/4};
```

We built temporal data objects and get the hazard rates in a single step:

```
tdc = TemporalData[c, {t}, ResamplingMethod → {"Interpolation", InterpolationOrder → 3}];
tdr = TemporalData[r, {t}, ResamplingMethod → {"Interpolation", InterpolationOrder → 3}];

haztd = TimeSeriesMap[#/(1 - R) &, tdc];
```

The term structure of hazard rates and corresponding CDS rates look as follows:

```
ListLinePlot[{tdc, haztd}, PlotTheme → "Web",
 PlotLegends → {"CDS", "Hazard Rate"},
 PlotLabel → "Term strucutre of CDS and Hazard rates"]
```



Once the hazard rates are known, we can value any CDS contract on a given reference entity. We define the premium and protection legs functions:

1) Premium leg:

```
premleg[x_, y_, n_] :=
 x * Sum[n * f * Exp[-(tdr["PathFunction"][i] + haztd["PathFunction"][i]) * i], {i, f, y, f}] // Quiet
```

2) Protection leg:

```
sevleg[y_, n_] :=
  (1 - R) *
   Sum[n * f * haztd["PathFunction"][y] *
    Exp[-(tdr["PathFunction"][i] + haztd["PathFunction"][i]) * i], {i, f, y, f}] // Quiet;
```

The fair-value CDS rate is then simply:

$$CDS = \frac{ProtectionLeg[T,1]}{PremiumLeg[1,T,1]}$$

```
CDS[y_] := sevleg[y, 1] / premleg[1, y, 1];
```

Which allows us to get all required CDS rates back:

```
cdscalc = {CDS[0.25], CDS[0.5], CDS[1], CDS[2], CDS[3], CDS[4], CDS[5]};
cdschart = TemporalData[cdscalc, {t}];
BarChart[cdschart, ChartStyle → "Rainbow", ChartLabels → t]
```



The calculated rates are exactly the same as the input, which shows that the hazard rates were correctly calibrated to the initial data.

```
(c - cdscalc) // Chop
{0, 0, 0, 0, 0, 0, 0}
```

The continuous CDS rates will then look as follows:

```
Plot[CDS[x], {x, 0, 5}, PlotTheme → "Business",
  PlotStyle → Red] // Quiet
```



Standardised DCS contracts have been modified and simplified to make their trading more transparent and easier. One of the features is the 'constant' premium coupon that market counterparties can apply to a broad group of reference entities. If such entity is of good credit standing, the premium is fixed at 1.00% p.a.

Since every entity represents different credit risk, the CDS rates in the market will all be different. This means that with the fixed premium rate, the value of the CDS contract will not be zero. This is resolved by the up-front payment settlement. Any deviation of the fair-value from the fixed premium is settled though this mechanics.

In our example with the given term structure, the 5Y CDS rate = 1.55%. This means that 1.00% payment will provide benefit o the premium payer and will have to be passed over the protection provider in form of up-front amount:

For the 5Y, 1 million contact this is around 25,000.

```
CDSVal[x_, y_, n_] := Chop[premleg[x, y, n] - sevleg[y, n]];

CDSVal[0.01, 5, 10^6]
-25021.3
```

The settlement amount is our case is an increasing function of time:

```
Table[{i, CDSVal[0.01, i, 10^6]}, {i, 5}];
ListLinePlot[%, PlotLabel → "Up-front amount", PlotStyle → Magenta]
```



Out of many structural variations, it is worth mentioning one – **Conditional CDS** or also known as CCDS. It is interesting variation where the protection is being offered on 'performance' of some other asset or contract. CCDS are frequently structured on derivative exposures – such as interest rate swaps or currency swaps. This enables the buyer to get protection on the unknown exposure amount in the future.

The valuation extension is trivial. Whilst the premium leg remains the same, the protection leg is adjusted to take into consideration the varying nature of the protection nominal.

Assume, that we transact a 5Y CCDS that provides protection on the 5Y non-standard IR swap (roller-coaster) with the following NPV profile:



We adjust the formula by bringing the variable expected nominal inside the summation term:

```
slegccds[y_, n_] :=
  (1 - R) * Sum[nomval[n, y]["PathFunction"][i] * f * haztd["PathFunction"][y] *
    Exp[-(tdr["PathFunction"][i] + haztd["PathFunction"][i]) * i], {i, f, y, f}] // Quiet;

CCDS[x_, y_, n_] := Chop[premleg[x, y, n] - slegccds[y, n]]
```

What is the fair premium for such contract? Simply CCDS protection leg divided by premium leg:
If the premium is paid on the fixed nominal = 1 mil. then the fair CCDS = 0.75%

```
ccdsrate = (slegccds[5, 10^6] / premleg[1, 5, 10^6]) * 10^4
74.9817
```